

HPEngErl v1.0

Software sobre los modelos Erlang-B, Erlang-C y Engset en HP-50g

Luís Díaz González
luisdecameixa@gmail.com

14 de julio de 2011

1. Introducción

Por todos es conocida la necesidad del uso de tablas en el cálculo relacionado con los modelos Erlang-B, Erlang-C y Engset. Estas tablas no cubren todos los valores posibles, un objetivo de este software es librarnos de esta limitación. El segundo objetivo es poder realizar el cálculo en cualquier sitio, es decir, no depender de un PC, la potencia de cálculo de la HP-50g permite esto.

Dado mi desconocimiento de SystemRPL y que tengo poco tiempo para aprenderlo, las funciones están escritas en UserRPL, desde luego que en SystemRPL su ejecución sería más rápida. Aún así he intentado llegar a un compromiso entre tiempo de ejecución y precisión en los cálculos. Las funciones que calculan la intensidad de tráfico son precisas hasta el cuarto decimal, para ello se utiliza el método de bisección. Otras funciones están basadas en su definición recursiva, que es un algoritmo eficiente y robusto para los cálculos.

HPEngErl v1.0 es distribuido bajo la licencia GPLv2, la cual puede consultarse en <http://www.gnu.org/licenses/gpl-2.0-standalone.html>.

2. Modelo Engset: M/M/m/m/N

2.1. Probabilidad de bloqueo. ENGS

$Eng(N, m, A) = ?$

Input:

3: intensidad A

2: servidores m

1: clientes N

Output:

1: probabilidad de bloqueo p

@ENGS

<< 0. RCLF -> A m N NA flags

<<-105 SF

IF 'm>0 AND N>0' THEN

IF m N < THEN

A ->NUM 'A' STO

IF A 0. >= THEN

IF A 0. !=

THEN

N A * 'NA' STO

0. 1. m

FOR K 1. + K NA K A * - / *

NEXT 1. + 1. SWAP /

ELSE

0.

END

IF m N == THEN 0. END

END

END

END

flags STOF

>>

>>

2.2. Cálculo del número de servidores. ENGSM

$$Eng(N, ?, A) < p$$

Input:

3: probabilidad de bloqueo p

2: intensidad A

1: clientes N

Output:

1: servidores m

@ENGSM

<< 0. 0. 0. RCLF -> E A N j s NA flags

<<-105 SF

A ->NUM 'A' STO

E ->NUM 'E' STO

IF 'E<1. AND E>0. AND N>0.' THEN

IF A 0. >= THEN

IF A 0. !=

THEN

N A * 'NA' STO

DO

1 'j' STO+

IF j N ==

THEN

0 's' STO

ELSE

1. s + j NA j A * - / * 's' STO

END

UNTIL 1. 1. s + / E <

END

IF j N < THEN j END

ELSE

0.

END

END

END

flags STOF

>>

>>

2.3. Cálculo de la intensidad de tráfico. ENGSA

$$Eng(N, m, ?) = p$$

Input:

3: probabilidad de bloqueo p

2: servidores m

1: clientes N

Output:

1: intensidad A

@ENGSA

```
<< 1. 0 0 0 0 0 RCLF
-> p m n a left testp mid midp right flags
<<-105 SF
IF 'p<1. AND p>0. AND m>0. AND n>0.' THEN
  IF m n <
  THEN
    p ->NUM 'p' STO
    a m n ENGS 'testp' STO
    WHILE testp p <
    REPEAT
      a 'left' STO
      2. 'a' STO*
      a m n ENGS 'testp' STO
    END

    a 'right' STO
    left right + 2. / 'mid' STO
    mid m n ENGS 'midp' STO

    WHILE right left - .00001 >
    REPEAT
      IF midp p <
      THEN
        mid 'left' STO
        left right + 2. / 'mid' STO
        mid m n ENGS 'midp' STO
      ELSE
        mid 'right' STO
        left right + 2. / 'mid' STO
        mid m n ENGS 'midp' STO
      END
    END
  END
  left
END
END
flags STOF
>>
>>
```

2.4. Cálculo del número de clientes. ENGSN

$$Eng(?, m, A) = p$$

Input:

3: probabilidad de bloqueo p

2: intensidad A

1: servidores m

Output:

1: clientes N

@ENGSN

<< 0. 0. 0. 0. 0. RCLF

-> E A m left right testp mid midp flags

<<-105 SF

A ->NUM 'A' STO

E ->NUM 'E' STO

IF 'E<1. AND E>0. AND m>0. AND A>0.' THEN

1 m + 'left' STO

32 m + 'right' STO

A m right ENGS 'testp' STO

WHILE testp E <

REPEAT

right 'left' STO

32. 'right' STO+

A m right ENGS 'testp' STO

END

WHILE right left - 1 >

REPEAT

left right + 2. / CEIL 'mid' STO

A m mid ENGS 'midp' STO

IF midp E <

THEN

mid 'left' STO

ELSE

mid 'right' STO

END

END

IF left m < THEN left END

END

flags STOF

>>

>>

3. Modelo Erlang-B: M/M/m/m

3.1. Probabilidad de bloqueo. ERB

$$E_B(m, A) = ?$$

Input:

2: intensidad A

1: servidores m

Output:

1: probabilidad de bloqueo p

@ERB

<< RCLF -> A m flags

<<-105 SF

IF m 0 > THEN

A ->NUM 'A' STO

IF A 0 >=

THEN

IF A 0. <

THEN

0. 1. m

FOR K 1. + K A / *

NEXT 1. + 1. SWAP /

ELSE

0.

END

END

END

flags STOF

>>

>>

3.2. Cálculo del número de servidores. ERBM

$$E_B(?, A) < p$$

Input:

2: probabilidad de bloqueo p

1: intensidad A

Output:

1: servidores m

@ERBM

<< 0. 0. RCLF -> E A j s flags

<<-105 SF

A ->NUM 'A' STO

E ->NUM 'E' STO

IF 'E<1. AND E>0.' THEN

IF A 0. >=

THEN

DO

1. 'j' STO+

1. s + j A / * 's' STO

UNTIL 1. 1. s + / E <

END

j

END

END

flags STOF

>>

>>

3.3. Cálculo de la intensidad de tráfico. ERBA

$$E_B(m, ?) = p$$

Input:

2: probabilidad de bloqueo p

1: servidores m

Output:

1: intensidad A

@ERBA

<< 1. 0. 0. 0. 0. 0. RCLF

-> p n a left testp mid midp right flags

<<-105 SF

p ->NUM 'p' STO

IF 'p<1. AND p>0. AND n>0' THEN

a n ERB 'testp' STO

WHILE testp p <

REPEAT

a 'left' STO

2. 'a' STO*

a n ERB 'testp' STO

END

a 'right' STO

left right + 2. / 'mid' STO

mid n ERB 'midp' STO

WHILE right left - .00001 >

REPEAT

IF midp p <

THEN

mid 'left' STO

left right + 2. / 'mid' STO

mid n ERB 'midp' STO

ELSE

mid 'right' STO

left right + 2. / 'mid' STO

mid n ERB 'midp' STO

END

END

left

END

flags STOF

>>

>>

4. Modelo Erlang-C: M/M/m

4.1. Probabilidad de espera. ERC

$$E_C(m, A) = ?$$

Input:

2: intensidad A

1: servidores m

Output:

1: probabilidad de espera p

@ERC

<< RCLF -> A m flags

<<-105 SF

IF m 0 > THEN

A ->NUM 'A' STO

IF 'A<=m AND A>0'

THEN

IF A 0. <

THEN

0. 1. m

FOR K 1. + K A / *

NEXT

1. + 1. SWAP / -> B

<<

m B * m A 1. B - * - /

>>

ELSE

0.

END

END

END

flags STOF

>>

>>

4.2. Cálculo del número de servidores. ERCM

$$E_C(?, A) < p$$

Input:

2: probabilidad de espera p

1: intensidad A

Output:

1: servidores m

@ERCM

<< 0. 0. 0. RCLF -> E A j s B flags

<<-105 SF

A ->NUM 'A' STO

E ->NUM 'E' STO

IF 'E<1. AND E>0.' THEN

IF A 0. >= THEN

DO

1. 'j' STO+

1. s + j A / * 's' STO

1. 1. s + / 'B' STO

UNTIL j B * j A 1. B - * - / E <

END

j

END

END

flags STOF

>>

>>

4.3. Cálculo de la intensidad de tráfico. ERCA

$$E_C(m, ?) = p$$

Input:

2: probabilidad de espera p

1: servidores m

Output:

1: intensidad A

@ERCA

<< 0. 0. 0. 0. 0. RCLF

-> p n left testp mid midp right flags

<<-105 SF

p ->NUM 'p' STO

IF 'p<1. AND p>0. AND n>0' THEN

n 'right' STO

left right + 2. / 'mid' STO

mid n ERC 'midp' STO

WHILE right left - .00001 >

REPEAT

IF midp p <

THEN

mid 'left' STO

left right + 2. / 'mid' STO

mid n ERC 'midp' STO

ELSE

mid 'right' STO

left right + 2. / 'mid' STO

mid n ERC 'midp' STO

END

END

left

END

flags STOF

>>

>>